
Construction of a video pattern generator

A pattern generator should be capable not only to produce the RGB components but also to
generate the corresponding composite video signal, with all its components: synchronism
pulses, luminance, modulated color subcarrier, etc.
The first thing we must define is the amount and type of patterns the set should generate,
because this will directly determine the characteristics and complexity of the circuit to
develop.
Our generator will be capable of producing four basic patterns:

It will also provide independent control for the three components, R, G and B, as well as for
luminance (Y) and chrominance (C). This fact greatly increases the pattern generation
capability, because the Raster could be of any of the eight colors, the Bars could be
monochrome or adopt different color combinations, as shown in the following diagrams:

As an additional control we can suppress the color burst, which is very useful to
troubleshoot the color processing stages of the TV receiver.
In order to select which of the four basic patterns will be generated we have two switches
(S4 y S5), selecting a pattern as show in the following table:



Switches: S4 OFF S4 ON

S5 OFF BARS RASTER

S5 ON CROSS-HATCH POINTS

The OFF (or “zero”) and ON (or “one”) states mean that the middle terminal of the switch is
connected to ground (0V) or VCC (5V) potentials respectively.
Once defined the characteristics of the generator let’s see how to do it.

Synchronism and pattern generation

Time base, synchronism and the four basic patterns generation will be accomplished by a
microcontroller (PIC16F84-10), so this stage would be basically a Software development.
At the end of this stage the microcontroller will be able to do the following:
• Generate a stable time base, from which all required times will be obtained.
• Generate in one of it pins, the one that corresponds to Bit 0 of PORTB, all the necessary

synchronism pulses to comply with the requirements of the selected television norm (N),
without adding video to this signal (pure synchronism signal).

• Generate the R, G and B signals, using three different pins. This signals will have the
information required to generate the selected pattern, and will not have added
synchronism pulses (pure video signal). The pin designation will be the following:

PORTB (2) = B (Blue)
PORTB (3) = R (Red)
PORTB (4) = G (Green)
(The corresponding bit of PORTB is indicated
between brackets)

• Accept in two of its pins, configured as inputs, the commands from switches S4 and S5,
in order to let the user select the pattern to generate. This two inputs correspond to two
Bits of PORTA, as shown:

PORTA (2) = S4
PORTA (3) = S5

Now we have the objectives clear; let’s see how the program works.

It is basically composed of four independent blocks, each of them has a complete set of
routines in order to generate a complete image. In the section of Diagrams I included a
flowchart of the program, which will help to understand the following explanation.
After a first stage, in which all variables are defined and loaded with the initial values, the
program reads the status of the two switches, S4 and S5. Depending on the combination of



these switches the program will be addressed to one of the four mentioned blocks,
corresponding to one of the four basic patterns.
Each one of these blocks begins with the generation of pre-equalizing pulses, then vertical
synchronism with the corresponding serrated pulses, followed by post-equalizing pulses.
After that the field is selected: odd or even. This is very important because we are working
with interlaced scanning, which means that the first line of the odd field is a full line, while
the even field begins with half line. If we did not take this into account the result would be
an unstable image, with a noticeable flicker in the upper side.
Note that in two of the four patterns (Cross-hatch and Points) we will use non-interlaced
scanning, in order to avoid the flicker of the fixed horizontal lines or points. In this case the
first line is always a full one; in order to compensate for this, we need to eliminate one pre-
equalizing pulse (half horizontal line), as previously shown in video signal diagrams.
Then the program generates 3 or 4 horizontal lines without video, depending on the field, so
as to compensate for time differences (only in interlaced patterns).
Now it is time to activate the RGB lines. After generating the horizontal synchronism and
back porch time, the video signal that correspond to the selected pattern is issues through
the RGB lines. How is this achieved?. Let’s see an example.
Suppose that we are generating a color bar signal. There are eight bars, so we must divide
the usable video time into eight equal intervals.
Before going on we must remember that the usable video time is the time during which the
generated information is effectively shown on the TV screen. In PAL-N each line has a
duration of 64 µsec.; this time includes 4.8 µsec. of horizontal synchronism, 1.9 µsec. of
front porch and 5 µsec. of back porch. So we only have 52.3 µsec. remaining to show video,
and this is our usable time.
Back to our example, we already defined the eight intervals. Let’s see how we should issue
the RGB signals in each of them:

As you can see this is a well known diagram. We already mentioned it to describe what a
simple RGB bar generator would do, and that is just what we are doing now.

Let’s analyze the generation of a Raster signal. This is much simpler: all the usable time we
must issue a high level signal through the three RGB lines. But, if all lines RGB are active at
the same time, the result will always be a white Raster. True. Color selection is achieved by
controlling the RGB signals outside the microcontroller, purely by hardware (three
switches).
To generate lines or points we need more elaborated routines, because we must control not
only the time in the horizontal direction but also the number of lines in the vertical direction,
in order to keep an equal distance between lines or points. However, this is not a problem;
we just add another variable to keep the count of lines and that’s all.
What about RGB lines?. They are all active when drawing lines or points, so they are white.

If you analyze the program you will probably notice that the generation of horizontal lines
and video signals within a video block is repeated three times. There is a simple reason for
this. In each run inside a video block a complete field is generated, that is 312.5 horizontal
lines. To achieve this we must count the lines and keep this count in a certain register. Since
I used an eight bit signed register, the maximum number that can be stored is 127, so I
needed to load it three times to achieve the required number of lines.

To finish with the video block, after each field is completed the program evaluates the
condition of the two switches (S4 and S5). If they remain unchanged, the program continues
within the same block; if there is any change it jumps to the initial evaluation routine, and
then goes to the selected video block.

And that’s all. May be there is still an unclear issue… how do I calculate the time inside the
program?. When you use a microcontroller this is very easy, you only have to count
“instruction cycles”. Using a 10 MHz oscillator and knowing the fact that each instruction
cycle needs four oscillator cycles, we can easily calculate the time of one instruction cycle:

Tosc = 1/fosc   
Tins = Tosc x 4
Tins = 1/10 MHz x 4 = 0.4 µsec.

If each instruction cycle lasts for 0.4 µsec., then we need to count 12 cycles in order to
obtain the horizontal synchronization pulse:

12 x 0.4 µsec. = 4.8 µsec.

In the same way we can calculate the cycles needed for a complete horizontal line, 160
instruction cycles:

160 x 0.4 µsec. = 64 µsec.

So, this is what the program does. It counts instructions and set or clear, as required, the Bit
0 of PORTB. In our case, during the equalization and synchronization pulses (H or V) this
bit will be clear (0V) and the rest of the time it will remain set (5V).



Composite Video generation

As we already stated, it is not enough to generate RGB signals to have a practical video
generator, that can be connected to TV receivers or VCRs. We must combine this RGB
signal with the synchronization signal and generate Composite Video, which is a practical
signal to test receivers.
We already analyzed all the steps needed to obtain Composite Video from RGB, so we will
not repeat it here. It is a hard process if you have to do it “manually”. Fortunately, there is
an integrated circuit, designed by Motorola®, that complies with the following
specifications:

• It has four signal inputs: Synchronism, R, G and B
• From RGB it generates luminance (Y)
• Has an in-circuit oscillator, which generates the color subcarrier
• Generates B-Y and R-Y signals, with the phase alternation required by the PAL system
• From B-Y and R-Y generates chrominance (C)
• Mixes Y with C to obtain Composite Video

As you can see, a single IC does exactly what we need. And it requires exactly the four
signal we already generated with the microcontroller.
This IC is the MC1377, RGB ENCODER, and with a few external components it can be
fully functional. In fact, I used the configuration suggested in the data sheet, with some
modifications to improve its performance.
The crystal used corresponds to PAL-N color subcarrier frequency, 3.582056 MHz. If you
want to use this equipment in Europe, in those countries using PAL-B/G/I, you only need to
replace the crystal by another one with the proper frequency (4.43 MHz) and make minor
adjustments to the TRIMMER CV1.
In this stage we have control of all the signals: RGB, Y, C and color burst. There are
basically six switches that derives the signal to ground, directly (RGB) or through a
capacitor (Y, C). In the case of color burst, to eliminate it, the switch (S8) disconnect a
capacitor (C04), responsible for generating the burst duration time.
Let’s see a summary of the switches and their function:

Switch Function

S1 G ON/OFF
S2 R ON/OFF
S3 B ON/OFF
S4 PROGRAM
S5 PROGRAM
S6 Y ON/OFF
S7 C ON/OFF
S8 BURST ON/OFF
S9 POWER



Once obtained the Composite Video signal, its level and impedance are adjusted by sending
it through a buffer circuit, composed by Q1, R14 and R15.

This concludes the signal generation, and practically the circuit description. I only have to
mention that the two main integrated circuits have different supply voltages, so you can see
a main power supply of 12V (8 AA alkaline batteries, this is a portable set) for the video
sector (U2 and Q1), and a secondary power supply of 5V, obtained from the main one, for
the microcontroller (U1).

Practical implementation of the video generator

In the following pages you will find all the necessary information, diagrams and drawings, to
construct a really working video generator. I have included the printed circuit board layout
in actual size, so you only need to print it on a transparency film and transfer to the board.
Note that the circuit is inverted, in order to make easier the mounting stage, using the
component layout diagram provided. In the actual board the text “Generador de video”
should be in the right direction.
About the program, I put it entirely in the final pages. You only need to copy it into a text
editor, assemble it and load it into the PIC, using the tools provided by Microchip® or the
ones you may have developed.

IMPORTANT ADVICE: While loading the program into the PIC, do not forget to set the
option for crystal (XT) operation. Otherwise, the crystal will not oscillate.

And that’s all. If everything is correctly placed the set will run as expected from the
beginning. The only adjust you may have to do is to move CV1 until you have a clear color
reproduction, which is quite simple.

I hope this project could be of use. I’ll be expecting your comments, suggestions and also
improvements you may think about. If you want a Spanish copy of this material do not
hesitate to contact me.

Marcelo F. Maggi - April, 1998
mmaggi@hotmail.com

R01 3K9 C13 0.1µ
R02 3K9 C14 220p
R03 3K9 C15 0.1µ
R04 1K C16 18p
R05 1K C17 150p
R06 1K C18 .02µ
R07 1K C19 100µ/16V
R08 1K C20 0.1µ
R09 1K C21 0.1µ
R10 68K C22 100µ/16V
R11 82K CV1 TRIMMER 5-45p
R12 10K D1 1N4007
R13 2K2 Q1 BF494C
R14 4K7 U1 PIC16F84-10
R15 2K7 U2 MC1377
R16 100 U3 LM78L05
C01 0.1µ X1 10.000MHz
C02 100µ/16V X2 3.582056MHz
C03 100µ/16V S1 2 POSITION SWITCH
C04 1500p S2 2 POSITION SWITCH
C05 10µ/25V S3 2 POSITION SWITCH
C06 10µ/25V S4 2 POSITION SWITCH
C07 10µ/25V S5 2 POSITION SWITCH
C08 .02µ S6 2 POSITION SWITCH
C09 .01µ S7 2 POSITION SWITCH
C10 15p S8 2 POSITION SWITCH
C11 15p S9 2 POSITION SWITCH
C12 0.1µ BT1 8 AA ALKALINE BATTERIES



5 cm





                                                                                

; ***** GENERADOR DE PATRONES PARA VIDEO *****
; VERSION 2.01
; GEN201.ASM
; (C) M. MAGGI - 30/08/1997
 
        list    p=16f84
 
;DEFINICION DE PUERTOS:
;PORTB(0): SYNC
;PORTB(2): AZUL
;PORTB(3): ROJO
;PORTB(4): VERDE
;NO USAR EL BIT 1 DEL PORTB
;PARA LOS PATRONES DE RASTER Y BARRAS EL VIDEO ES ENTRELAZADO
;LOS PUNTOS Y EL CROSSHATCH SE HACEN CON VIDEO NO ENTRELAZADO PARA EVITAR EL
;"FLICKER"
 
CBLOCK  0X0C            ;VARIABLES
        DURHOR,CANTHB1,CANTHB2,BLKLIN,CANTPRE,DUREQU,CANTVER,DURVER,CANTPOS
        TIEMPO,FIELD,CARRY
        WHITE,YELLOW,CYAN,GREEN,MAGEN,RED,BLUE,BLACK,CANTLIN
ENDC
 
PORTA   EQU     5
TRISA   EQU     85H
PORTB   EQU     6 
TRISB   EQU     86H
STATUS  EQU     3
RP0     EQU     5
BLANCO  EQU     B'00011101'
AMARIL  EQU     B'00011001'
CYANO   EQU     B'00010101'
VERDE   EQU     B'00010001'
MAGENT  EQU     B'00001101'
ROJO    EQU     B'00001001'
AZUL    EQU     B'00000101'
NEGRO   EQU     B'00000001'
;
        CLRF    PORTA           ;TODOS LOS BITS EN 0
        CLRF    PORTB           ;TODOS LOS BITS EN 0
        BSF     STATUS,RP0      ;SELECCIONA BANCO DE REGISTROS 1   
        MOVLW   B'11111111'
        MOVWF   TRISA           ;TODOS LOS BITS DEL PUERTO A COMO ENTRADAS
        CLRF    TRISB^80H       ;TODOS LOS BITS DEL PUERTO B COMO SALIDA
        BCF     STATUS,RP0      ;SELECCIONA BANCO DE REGISTROS O 
;
        MOVLW   0
        MOVWF   CARRY           ;VARIABLE CONTROLAR EL ESTADO DEL CARRY
        RRF     CARRY           ;CARRY FLAG A "0"
        MOVLW   B'10101010'
        MOVWF   FIELD           ;CONTROL DEL CAMPO

LECTURA BTFSS   PORTA,3         ;SE LEE EL TECLADO
        GOTO    LECT1           ;SE USAN LOS BITS 2 Y 3 DEL PUERTO A
        BTFSC   PORTA,2         ;FUNCION:       BIT3    BIT2
        GOTO    INICIO3         ;BARRAS         0       0
        GOTO    INICIO2         ;RASTER         0       1
LECT1   BTFSC   PORTA,2         ;CROSSHATCH     1       0
        GOTO    INICIO1         ;PUNTOS         1       1

;***** BARRAS DE COLOR *****

INICIO  RRF     FIELD           ;CARRY PASA AL BIT 7 DE FIELD, BIT 0 AL CARRY
        MOVLW   D'3'            ;LINEAS SIN VIDEO LUEGO DE LA POSECUALIZACION

                                                                                



                                                                                

        BTFSS   FIELD,0         ;SI ES EL CAMPO 1 SE HACEN SOLO 3 LINEAS
        MOVLW   D'4'            ;4 LINEAS EN EL CAMPO 2
        MOVWF   BLKLIN
        MOVLW   D'99'
        MOVWF   CANTHB1         ;CANTIDAD DE LINEAS HORIZONTALES EN UN BLOQUE
        MOVLW   D'3'
        MOVWF   CANTHB2         ;CANTIDAD DE BLOQUES (3)
        MOVLW   5
        MOVWF   CANTPRE         ;PULSOS DE PREECUALIZACION
        MOVLW   5
        MOVWF   CANTVER         ;PULSOS DE SINCRONISMO VERTICAL
        MOVLW   5
        MOVWF   CANTPOS         ;PULSOS DE POSECUALIZACION
   
PREEQU  BCF     PORTB,0         ;DURACION: 2,6µS ABAJO
        MOVLW   D'23'
        MOVWF   DUREQU
        NOP
        NOP
        NOP
        BSF     PORTB,0
LOOP1   DECFSZ  DUREQU          ;SE COMPLETAN LOS 32µS ARRIBA
        GOTO    LOOP1
        NOP
        NOP
        DECFSZ  CANTPRE
        GOTO    PREEQU
        NOP
VERT    BCF     PORTB,0
        MOVLW   D'22'
        MOVWF   DURVER
LOOP2   DECFSZ  DURVER
        GOTO    LOOP2   
        BSF     PORTB,0         ;DURACION: 4.8µS ARRIBA ("SERRATED PULSES")
        MOVLW   2
        MOVWF   TIEMPO
TIME    DECFSZ  TIEMPO
        GOTO    TIME
        NOP
        DECFSZ  CANTVER
        GOTO    VERT
        NOP
POSEQU  BCF     PORTB,0
        MOVLW   D'23'
        MOVWF   DUREQU
        NOP
        NOP
        NOP
        BSF     PORTB,0
LOOP3   DECFSZ  DUREQU
        GOTO    LOOP3
        NOP
        NOP
        DECFSZ  CANTPOS
        GOTO    POSEQU
        NOP
 
;SE EMPIEZAN A BARRER LAS LINEAS HORIZONTALES
;LA PRIMERA LINEA ES COMPLETA EN EL CAMPO 1, EN TANTO QUE ES SOLO MEDIA LINEA
;EN EL CAMPO 2, Y NO COMIENZA CON UN PULSO DE SINCRONISMO

        RLF     PORTB           ;1 O 1/2 LINEA H SEGUN EL CAMPO
        NOP                     ;SE PASA EL CARRY AL BIT 0 DEL PUERTO B
        NOP                     ;CAMPO 1: 1 LINEA Y PULSO DE SINC (CARRY=0)

                                                                                



                                                                                

        NOP                     ;CAMPO 2: 1/2 LINEA SIN PULSO DE SINC (CARRY=1)
        NOP
        NOP
        NOP
        NOP
        MOVLW   D'21'           ;TIEMPO PARA 1/2 H (80 CICLOS TOTAL)
        BTFSS   PORTB,0         ;SI HAY H SYNC (CAMPO 1) SE AGREGA MAS TIEMPO
        ADDLW   D'27'           ;TIEMPO PARA 1 H (160 CICLOS TOTAL)
        MOVWF   DURHOR
        BSF     PORTB,0
        BTFSS   FIELD,0
        GOTO    NEXT            ;SE PIERDE 1 CICLO MAS (SOLO 1/2 H)
NEXT    BCF     PORTB,1
        NOP
LOOP    DECFSZ  DURHOR
        GOTO    LOOP
        NOP

;SE HACEN 3 O 4 LINEAS EN BLANCO PARA CUMPLIR CON LAS 625 LINEAS DE LA NORMA N
;SI ES EL CAMPO 1 SE HACEN SOLO 3 LINEAS, YA QUE ANTES SE HIZO 1 DE MAS

HORIZ   BCF     PORTB,0           
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
TIME3   DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    TIME3
        NOP
        NOP
        MOVLW   D'48'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
LOOPH3  DECFSZ  DURHOR
        GOTO    LOOPH3
        NOP
        DECFSZ  BLKLIN
        GOTO    HORIZ
        NOP     

;SE HACEN 3 BLOQUES DE 99 LINEAS HORIZONTALES
;3*(99+1)=300 LINEAS

HORIZ1  BCF     PORTB,0         
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
TIME1   DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO TIME1
        NOP
        NOP
        MOVLW   D'31'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
        NOP
        NOP
        NOP
        MOVLW   5
        MOVWF   WHITE
        MOVWF   YELLOW
        MOVWF   CYAN
        MOVWF   GREEN
        MOVWF   MAGEN
        MOVWF   RED
        MOVWF   BLUE
        MOVWF   BLACK
        MOVLW   BLANCO

                                                                                



                                                                                

        MOVWF   PORTB   
WHITE1  DECFSZ  WHITE
        GOTO    WHITE1
        MOVLW   AMARIL
        MOVWF   PORTB   
YELLO1  DECFSZ  YELLOW
        GOTO    YELLO1
        MOVLW   CYANO
        MOVWF   PORTB   
CYAN1   DECFSZ  CYAN
        GOTO    CYAN1
        MOVLW   VERDE
        MOVWF   PORTB   
GREEN1  DECFSZ  GREEN
        GOTO    GREEN1
        MOVLW   MAGENT
        MOVWF   PORTB   
MAGEN1  DECFSZ  MAGEN
        GOTO    MAGEN1
        MOVLW   ROJO
        MOVWF   PORTB   
RED1    DECFSZ  RED
        GOTO    RED1
        MOVLW   AZUL
        MOVWF   PORTB   
BLUE1   DECFSZ  BLUE
        GOTO    BLUE1
        MOVLW   NEGRO
        MOVWF   PORTB   
BLACK1  DECFSZ  BLACK
        GOTO    BLACK1
        NOP
        NOP
        NOP
        NOP
        DECFSZ  CANTHB1
        GOTO    HORIZ1
        NOP

HORIZ2  BCF     PORTB,0         ;BIT 0 BAJO
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
TIME2   DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    TIME2
        MOVLW   D'99'
        MOVWF   CANTHB1
        MOVLW   D'31'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
        NOP
        NOP
        NOP
        MOVLW   5
        MOVWF   WHITE
        MOVWF   YELLOW
        MOVWF   CYAN
        MOVWF   GREEN
        MOVWF   MAGEN
        MOVWF   RED
        MOVWF   BLUE
        MOVWF   BLACK
        MOVLW   BLANCO
        MOVWF   PORTB   
WHITE2  DECFSZ  WHITE

                                                                                



                                                                                

        GOTO    WHITE2
        MOVLW   AMARIL
        MOVWF   PORTB   
YELLO2  DECFSZ  YELLOW
        GOTO    YELLO2
        MOVLW   CYANO
        MOVWF   PORTB   
CYAN2   DECFSZ  CYAN
        GOTO    CYAN2
        MOVLW   VERDE
        MOVWF   PORTB   
GREEN2  DECFSZ  GREEN
        GOTO    GREEN2
        MOVLW   MAGENT
        MOVWF   PORTB   
MAGEN2  DECFSZ  MAGEN
        GOTO    MAGEN2
        MOVLW   ROJO
        MOVWF   PORTB   
RED2    DECFSZ  RED
        GOTO    RED2
        MOVLW   AZUL
        MOVWF   PORTB   
BLUE2   DECFSZ  BLUE
        GOTO    BLUE2
        MOVLW   NEGRO
        MOVWF   PORTB   
BLACK2  DECFSZ  BLACK
        GOTO    BLACK2
        NOP
        NOP
        NOP
        NOP
        DECFSZ  CANTHB2
        GOTO    HORIZ1
        NOP
 
;ESTA ULTIMA LINEA/MEDIA LINEA, LA 305, LA USO PARA CARGAR VARIABLES        
 
        BCF     PORTB,0         ;BIT 0 PASA A NIVEL BAJO
        NOP                     ;PIERDO TIEMPO PARA
        NOP                     ;HACER LOS 4,8µS
        MOVLW   0               ;NO USAR EL BIT 1 DEL PORTB, BIT 0 = SYNC
        BTFSC   FIELD,0
        MOVLW   1
        MOVWF   CARRY
        NOP
        MOVLW   D'15'
        BTFSC   FIELD,0
        ADDLW   D'24'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 PASA A NIVEL ALTO
        BTFSS   FIELD,0
        GOTO    NEXT1
NEXT1   NOP
        NOP
LOOPH5  DECFSZ  DURHOR
        GOTO    LOOPH5
        RRF     CARRY           ;CARRY = 1 SI 1 H, CARRY = 0 SI 1/2 H
        BTFSC   PORTA,2
        GOTO    LECTURA
        BTFSC   PORTA,3
        GOTO    LECTURA
        NOP

                                                                                



                                                                                

        NOP
        NOP
        NOP
        NOP
        GOTO    INICIO

;***** RASTER *****

INICIO1 RRF     FIELD
        MOVLW   D'3'
        BTFSS   FIELD,0
        MOVLW   D'4'
        MOVWF   BLKLIN
        MOVLW   D'99'
        MOVWF   CANTHB1
        MOVLW   D'3'
        MOVWF   CANTHB2
        MOVLW   5
        MOVWF   CANTPRE
        MOVLW   5
        MOVWF   CANTVER
        MOVLW   5
        MOVWF   CANTPOS
   
APREEQU BCF     PORTB,0         ;DURACION: 2,6µS ABAJO
        MOVLW   D'23'
        MOVWF   DUREQU
        NOP
        NOP
        NOP
        BSF     PORTB,0
ALOOP1  DECFSZ  DUREQU          ;SE COMPLETAN LOS 32µS ARRIBA
        GOTO    ALOOP1
        NOP
        NOP
        DECFSZ  CANTPRE
        GOTO    APREEQU
        NOP
AVERT   BCF     PORTB,0
        MOVLW   D'22'
        MOVWF   DURVER
ALOOP2  DECFSZ  DURVER
        GOTO    ALOOP2   
        BSF     PORTB,0         ;DURACION: 4.8µS ARRIBA ("SERRATED PULSES")
        MOVLW   2
        MOVWF   TIEMPO
ATIME   DECFSZ  TIEMPO
        GOTO    ATIME
        NOP
        DECFSZ  CANTVER
        GOTO    AVERT
        NOP
APOSEQU BCF     PORTB,0
        MOVLW   D'23'
        MOVWF   DUREQU
        NOP
        NOP
        NOP
        BSF     PORTB,0
ALOOP3  DECFSZ  DUREQU
        GOTO    ALOOP3
        NOP
        NOP
        DECFSZ  CANTPOS

                                                                                



                                                                                

        GOTO    APOSEQU
        NOP
 
        RLF     PORTB           ;1 O 1/2 LINEA H SEGUN EL CAMPO
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        MOVLW   D'21'           ;TIEMPO PARA 1/2 H (80 CICLOS TOTAL)
        BTFSS   PORTB,0
        ADDLW   D'27'           ;TIEMPO PARA 1 H (160 CICLOS TOTAL)
        MOVWF   DURHOR
        BSF     PORTB,0
        BTFSS   FIELD,0
        GOTO    ANEXT           ;SE PIERDE 1 CICLO MAS (SOLO 1/2 H)
ANEXT   BCF     PORTB,1
        NOP
ALOOP   DECFSZ  DURHOR
        GOTO    ALOOP
        NOP

AHORIZ  BCF     PORTB,0           
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
ATIME3  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    ATIME3
        NOP
        NOP
        MOVLW   D'48'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
ALOOPH3 DECFSZ  DURHOR
        GOTO    ALOOPH3
        NOP
        DECFSZ  BLKLIN
        GOTO    AHORIZ
        NOP     

AHORIZ1 BCF     PORTB,0         
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
ATIME1  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    ATIME1
        NOP
        NOP
        MOVLW   D'44'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        MOVLW   B'00011101'
        MOVWF   PORTB
ALOOPH4 DECFSZ  DURHOR
        GOTO    ALOOPH4

                                                                                



                                                                                

        MOVLW   B'00000001'
        MOVWF   PORTB
        DECFSZ  CANTHB1
        GOTO    AHORIZ1
        NOP

        BCF     PORTB,0         
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
ATIME2  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    ATIME2
        MOVLW   D'99'
        MOVWF   CANTHB1
        MOVLW   D'44'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        MOVLW   B'00011101'
        MOVWF   PORTB
ALOOPH5 DECFSZ  DURHOR
        GOTO    ALOOPH5
        MOVLW   B'00000001'
        MOVWF   PORTB
        DECFSZ  CANTHB2
        GOTO    AHORIZ1
        NOP

;ESTA ULTIMA LINEA/MEDIA LINEA, LA 305, LA USO PARA CARGAR VARIABLES        
 
        BCF     PORTB,0         ;BIT 0 PASA A NIVEL BAJO
        NOP                     ;PIERDO TIEMPO PARA
        NOP                     ;HACER LOS 4,8µS
        MOVLW   0               ;NO USAR EL BIT 1 DEL PORTB, BIT 0 = SYNC
        BTFSC   FIELD,0
        MOVLW   1
        MOVWF   CARRY
        NOP
        MOVLW   D'15'
        BTFSC   FIELD,0
        ADDLW   D'24'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 PASA A NIVEL ALTO
        BTFSS   FIELD,0
        GOTO    ANEXT1
ANEXT1  NOP
        NOP
ALOOPH6 DECFSZ  DURHOR
        GOTO    ALOOPH6
        RRF     CARRY           ;CARRY = 1 SI 1 H, CARRY = 0 SI 1/2 H
        BTFSS   PORTA,2
        GOTO    LECTURA
        BTFSC   PORTA,3
        GOTO    LECTURA
        NOP
        NOP
        NOP

                                                                                



                                                                                

        NOP
        NOP
        GOTO    INICIO1

;***** CROSSHATCH *****

INICIO2 RRF     FIELD
        NOP
        NOP
        MOVLW   D'4'
        MOVWF   BLKLIN
        MOVLW   D'28'
        MOVWF   CANTHB1
        MOVLW   D'10'
        MOVWF   CANTHB2
        MOVLW   4
        MOVWF   CANTPRE         ;SOLO 4 PULSOS POR SER VIDEO NO ENTRELAZADO
        MOVLW   5
        MOVWF   CANTVER
        MOVLW   5
        MOVWF   CANTPOS
   
BPREEQU BCF     PORTB,0         ;DURACION: 2,6µS ABAJO
        MOVLW   D'23'
        MOVWF   DUREQU
        NOP
        NOP
        NOP
        BSF     PORTB,0
BLOOP1  DECFSZ  DUREQU          ;SE COMPLETAN LOS 32µS ARRIBA
        GOTO    BLOOP1
        NOP
        NOP
        DECFSZ  CANTPRE
        GOTO    BPREEQU
        NOP
BVERT   BCF     PORTB,0
        MOVLW   D'22'
        MOVWF   DURVER
BLOOP2  DECFSZ  DURVER
        GOTO    BLOOP2   
        BSF     PORTB,0         ;DURACION: 4.8µS ARRIBA ("SERRATED PULSES")
        MOVLW   2
        MOVWF   TIEMPO
BTIME   DECFSZ  TIEMPO
        GOTO    BTIME
        NOP
        DECFSZ  CANTVER
        GOTO    BVERT
        NOP
BPOSEQU BCF     PORTB,0
        MOVLW   D'23'
        MOVWF   DUREQU
        NOP
        NOP
        NOP
        BSF     PORTB,0
BLOOP3  DECFSZ  DUREQU
        GOTO    BLOOP3
        NOP
        NOP
        DECFSZ  CANTPOS
        GOTO    BPOSEQU
        NOP

                                                                                



                                                                                

 
        NOP                     ;1/2 LINEA H (NO ENTRELAZADO)
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        MOVLW   D'21'           ;TIEMPO PARA 1/2 H (80 CICLOS TOTAL)
        NOP
        NOP
        MOVWF   DURHOR
        NOP
        NOP
        NOP
        NOP
        BCF     PORTB,1
        NOP
BLOOP   DECFSZ  DURHOR
        GOTO    BLOOP
        NOP

BHORIZ  BCF     PORTB,0           
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
BTIME3  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    BTIME3
        NOP
        NOP
        MOVLW   D'48'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
BLOOPH3 DECFSZ  DURHOR
        GOTO    BLOOPH3
        NOP
        DECFSZ  BLKLIN
        GOTO    BHORIZ
        NOP     

BHORIZ1 BCF     PORTB,0         
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
BTIME1  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    BTIME1
        NOP
        NOP
        MOVLW   9
        MOVWF   CANTLIN
        BSF     PORTB,0         ;BIT 0 ALTO
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
BLOOPHA MOVLW   B'00011100'
        ADDWF   PORTB
        SUBWF   PORTB
        NOP
        MOVLW   2

                                                                                



                                                                                

        MOVWF   DURHOR
BLOOPH4 DECFSZ  DURHOR
        GOTO    BLOOPH4
        DECFSZ  CANTLIN
        GOTO    BLOOPHA
        NOP
        MOVLW   B'00011100'
        ADDWF   PORTB
        SUBWF   PORTB
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        DECFSZ  CANTHB1
        GOTO    BHORIZ1
        NOP

        BCF     PORTB,0         
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
BTIMEZ  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    BTIMEZ
        NOP
        NOP
        MOVLW   D'44'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        MOVLW   B'00011101'
        MOVWF   PORTB
BLOOPHZ DECFSZ  DURHOR
        GOTO    BLOOPHZ
        MOVLW   B'00000001'
        MOVWF   PORTB
        NOP
        NOP
        NOP
        
        BCF     PORTB,0         
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
BTIME2  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    BTIME2
        MOVLW   D'28'
        MOVWF   CANTHB1
        MOVLW   D'44'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP

                                                                                



                                                                                

        NOP
        NOP
        NOP
        MOVLW   B'00011101'
        MOVWF   PORTB
BLOOPH5 DECFSZ  DURHOR
        GOTO    BLOOPH5
        MOVLW   B'00000001'
        MOVWF   PORTB
        DECFSZ  CANTHB2
        GOTO    BHORIZ1
        NOP

;ESTA ULTIMA MEDIA LINEA, LA 305, LA USO PARA CARGAR VARIABLES        
 
        BCF     PORTB,0         ;BIT 0 PASA A NIVEL BAJO
        NOP                     ;PIERDO TIEMPO PARA
        NOP                     ;HACER LOS 4,8µS
        MOVLW   0               ;NO USAR EL BIT 1 DEL PORTB, BIT 0 = SYNC
        BTFSC   FIELD,0
        MOVLW   1
        MOVWF   CARRY
        NOP
        MOVLW   D'15'
        NOP
        NOP
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 PASA A NIVEL ALTO
        NOP
        NOP
        NOP
        NOP
        NOP
BLOOPH6 DECFSZ  DURHOR
        GOTO    BLOOPH6
        RRF     CARRY           ;CARRY = 1 SI 1 H, CARRY = 0 SI 1/2 H
        BTFSC   PORTA,2
        GOTO    LECTURA
        BTFSS   PORTA,3
        GOTO    LECTURA
        NOP
        NOP
        NOP
        NOP
        NOP
        GOTO    INICIO2

;***** PUNTOS *****

INICIO3 RRF     FIELD
        NOP
        NOP
        MOVLW   D'4'
        MOVWF   BLKLIN
        MOVLW   D'28'
        MOVWF   CANTHB1
        MOVLW   D'10'
        MOVWF   CANTHB2
        MOVLW   4
        MOVWF   CANTPRE
        MOVLW   5
        MOVWF   CANTVER
        MOVLW   5
        MOVWF   CANTPOS

                                                                                



                                                                                

   
CPREEQU BCF     PORTB,0         ;DURACION: 2,6µS ABAJO
        MOVLW   D'23'
        MOVWF   DUREQU
        NOP
        NOP
        NOP
        BSF     PORTB,0
CLOOP1  DECFSZ  DUREQU          ;SE COMPLETAN LOS 32µS ARRIBA
        GOTO    CLOOP1
        NOP
        NOP
        DECFSZ  CANTPRE
        GOTO    CPREEQU
        NOP
CVERT   BCF     PORTB,0
        MOVLW   D'22'
        MOVWF   DURVER
CLOOP2  DECFSZ  DURVER
        GOTO    CLOOP2   
        BSF     PORTB,0         ;DURACION: 4.8µS ARRIBA ("SERRATED PULSES")
        MOVLW   2
        MOVWF   TIEMPO
CTIME   DECFSZ  TIEMPO
        GOTO    CTIME
        NOP
        DECFSZ  CANTVER
        GOTO    CVERT
        NOP
CPOSEQU BCF     PORTB,0
        MOVLW   D'23'
        MOVWF   DUREQU
        NOP
        NOP
        NOP
        BSF     PORTB,0
CLOOP3  DECFSZ  DUREQU
        GOTO    CLOOP3
        NOP
        NOP
        DECFSZ  CANTPOS
        GOTO    CPOSEQU
        NOP
 
        NOP                     ;1/2 LINEA H
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        MOVLW   D'21'           ;TIEMPO PARA 1/2 H (80 CICLOS TOTAL)
        NOP
        NOP
        MOVWF   DURHOR
        NOP
        NOP
        NOP
        NOP
        BCF     PORTB,1
        NOP
CLOOP   DECFSZ  DURHOR
        GOTO    CLOOP

                                                                                



                                                                                

        NOP

CHORIZ  BCF     PORTB,0           
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
CTIME3  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    CTIME3
        NOP
        NOP
        MOVLW   D'48'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
CLOOPH3 DECFSZ  DURHOR
        GOTO    CLOOPH3
        NOP
        DECFSZ  BLKLIN
        GOTO    CHORIZ
        NOP     

CHORIZ1 BCF     PORTB,0         
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
CTIME1  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    CTIME1
        NOP
        NOP
        MOVLW   D'48'
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 ALTO
CLOOPHZ DECFSZ  DURHOR
        GOTO    CLOOPHZ
        NOP
        DECFSZ  CANTHB1
        GOTO    CHORIZ1
        NOP
        
        BCF     PORTB,0         
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
CTIMEZ  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    CTIMEZ
        NOP
        NOP
        MOVLW   9
        MOVWF   CANTLIN
        BSF     PORTB,0         ;BIT 0 ALTO
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
CLOOPHA MOVLW   B'00011100'
        ADDWF   PORTB
        SUBWF   PORTB
        NOP
        MOVLW   2
        MOVWF   DURHOR
CLOOPH4 DECFSZ  DURHOR
        GOTO    CLOOPH4
        DECFSZ  CANTLIN

                                                                                



                                                                                

        GOTO    CLOOPHA
        NOP
        MOVLW   B'00011100'
        ADDWF   PORTB
        SUBWF   PORTB
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP

        BCF     PORTB,0         
        MOVLW   2
        MOVWF   TIEMPO          ;PIERDO TIEMPO PARA
CTIME2  DECFSZ  TIEMPO          ;HACER LOS 4,8µS
        GOTO    CTIME2
        MOVLW   D'28'
        MOVWF   CANTHB1
        MOVLW   9
        MOVWF   CANTLIN
        BSF     PORTB,0         ;BIT 0 ALTO
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
CLOOPHB MOVLW   B'00011100'
        ADDWF   PORTB
        SUBWF   PORTB
        NOP
        MOVLW   2
        MOVWF   DURHOR
CLOOPH5 DECFSZ  DURHOR
        GOTO    CLOOPH5
        DECFSZ  CANTLIN
        GOTO    CLOOPHB
        NOP
        MOVLW   B'00011100'
        ADDWF   PORTB
        SUBWF   PORTB
        NOP
        NOP
        NOP
        NOP
        NOP
        NOP
        DECFSZ  CANTHB2
        GOTO    CHORIZ1
        NOP

;ESTA ULTIMA MEDIA LINEA, LA 305, LA USO PARA CARGAR VARIABLES        
 
        BCF     PORTB,0         ;BIT 0 PASA A NIVEL BAJO
        NOP                     ;PIERDO TIEMPO PARA
        NOP                     ;HACER LOS 4,8µS
        MOVLW   0               ;NO USAR EL BIT 1 DEL PORTB, BIT 0 = SYNC

                                                                                



                                                                                

        BTFSC   FIELD,0
        MOVLW   1
        MOVWF   CARRY
        NOP
        MOVLW   D'15'
        NOP
        NOP
        MOVWF   DURHOR
        BSF     PORTB,0         ;BIT 0 PASA A NIVEL ALTO
        NOP
        NOP
        NOP
        NOP
        NOP
CLOOPH6 DECFSZ  DURHOR
        GOTO    CLOOPH6
        RRF     CARRY           ;CARRY = 1 SI 1 H, CARRY = 0 SI 1/2 H
        BTFSS   PORTA,2
        GOTO    LECTURA
        BTFSS   PORTA,3
        GOTO    LECTURA
        NOP
        NOP
        NOP
        NOP
        NOP
        GOTO    INICIO3
        
        END

